
Bessel_6P_Group_Delay

.OPTIONS  GMIN=1e-18    METHOD=euler   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
VREF      VREF    0      PULSE( -.5 .5 1u 1u  1u  10m 20m )
.include Bessel_6P_VCF_1.txt
.include Bessel_6P_VCF_100m.txt 
.include Bessel_6P_VCF_30m.txt     
Vtime    Vtime  0      PWL ( 0 0 40m 12.566370614359172)
BIN      VIN    0      v=    v(VA)*v(OUTA) + v(VB)*v(OUTB) + v(VC)*v(OUTC)
B1       OUT1   0      v=    v(V1S)*sin(1*v(Vtime))+ v(V1C)*cos(1*v(Vtime))
B3       OUT3   0      v=    v(V3S)*sin(3*v(Vtime))+ v(V3C)*cos(3*v(Vtime))
B5       OUT5   0      v=    v(V5S)*sin(5*v(Vtime))+ v(V5C)*cos(5*v(Vtime))
B7       OUT7   0      v=    v(V7S)*sin(7*v(Vtime))+ v(V7C)*cos(7*v(Vtime))
B9       OUT9   0      v=    v(V9S)*sin(9*v(Vtime))+ v(V9C)*cos(9*v(Vtime))
B11      OUT11  0      v=    v(V11S)*sin(11*v(Vtime))+ v(V11C)*cos(11*v(Vtime))
BALL     OUTALL 0      v=    v(OUT1)+ v(OUT3)+ v(OUT5)+ v(OUT7) +v(OUT9) +v(OUT11)
V1S      V1S    0      DC    .5
V1C      V1C    0      DC    .5   
V3S      V3S    0      DC    .5
V3C      V3C    0      DC    .5  
V5S      V5S    0      DC    .5
V5C      V5C    0      DC    .5    
V7S      V7S    0      DC    .5
V7C      V7C    0      DC    .5   
V9S      V9S    0      DC    .5
V9C      V9C    0      DC    .5 
V11S     V11S   0      DC    .5
V11C     V11C   0      DC    .5  

VA       VA     0      DC    1     
VB       VB     0      DC    0
VC       VC     0      DC    0   

.control
set         pensize = 2
foreach   PWLNumb   0 1 2
if          ($PWLNumb = 0)
alter        va dc = 1
endif
if          ($PWLNumb = 1)
alter        va dc = 0
alter        vb dc = 1
endif
if          ($PWLNumb = 2)
alter        vb dc = 0
alter        vc dc = 1
endif

tran        .05m     40m      0  .05m
linearize
set         specwindow = "rectangular"
spec        25      1000     25     v(vin)  

set    s1= im(vin[1])
set    c1= real(vin[1])
set    s3= im(vin[5])
set    c3= real(vin[5])
set    s5= im(vin[9])
set    c5= real(vin[9])
set    s7= im(vin[13])
set    c7= real(vin[13])
set    s9= im(vin[17])
set    c9= real(vin[17])
set    s11= im(vin[21])
set    c11= real(vin[21])
alter  v1s dc = $s1
alter  v1c dc = $c1
alter  v3s dc = $s3
alter  v3c dc = $c3
alter  v5s dc = $s5
alter  v5c dc = $c5
alter  v7s dc = $s7
alter  v7c dc = $c7
alter  v9s dc = $s9
alter  v9c dc = $c9
alter  v11s dc = $s11
alter  v11c dc = $c11
tran      .05m     40m      0  .05m
plot      vin  out1 out3 out5 out7 out9 out11 outall  1.1*vref
plot      vin  out1 out3 out5 out7 out9 out11 outall  1.1*vref xlimit 20m 24m
 



end
.endc
.end

=====================END_OF_SPICE=======================

A good way to see how the Bessel provides for Low Phase 
Distortion is to view what is happening to a input
square wave's harmonics in a bessel filter.

The vin signals for the graph below come from the
Voltage Controlled filter simulations. This signal's 
odd harmonics are ploted up to the 11th harmonic.
The fundamental and these harmonics are then summed
together to provide a outall signal. The reference
square wave that went into the VCF is shown in yellow.

Obviously 11 harmonics are not enough to recontruct the
vin signal. In this case the 3dB bandwidth was at the 
40th harmonic. The delay of vin and all its harmonics
needs to be looked at with higher resolution.



All components of the vin signal are time delayed from
the reference signal by the same 300usec.



When the bessel's frequency response is reduced by a 
factor of ten, there is now about a 2.7msec delay in
all components of vin. Only the third and fifth 
harmonics are now present. 



Finer resolution reveals that the fundamental and all it's
harmonics are being shifted by the same 2.7msec time delay.
Hence the phase relationship between the fundamental and
all it's harmonics is conserved without any phase distortion.
Another term is group delay which can be seen that the 
whole signal as group of frequencies have be delayed by
the same amount. This does not apply to a butterworth
filter.


